Using SAS® to perform a stratified t-test

John Salter, Oxford Pharmaceutical Sciences Ltd, Oxford, UK

ABSTRACT
This paper examines the stratified t-test approach and provides some SAS® code for provision of simple outputs as well as detailed datasets for output into tables and statistical appendices.

INTRODUCTION
The stratified t-test is an often overlooked method of analysis that provides a robust test of comparison. Nowadays, with the advent of greater computational power and faster processing speeds, other as powerful tests tend to be used instead – a straightforward analysis of covariance (ANCOVA) will provide similar output, but without the detailed breakdown of strata detail. However, if a stratified t-test has been pre-specified in a Reporting and Analysis Plan (RAP), then we need to consider an approach as detailed below.

This paper examines a “back-to-basics” approach to deriving the stratified t-test, with an aim to providing a detailed breakdown of the analysis itself, for provision in statistical tables and appendices. It will also examine the differences between the output from the stratified t-test and an analysis of covariance via simulation, to provide more insight into the use of either method.

METHODOLOGY
The test statistic for the pairwise comparison of test vs. active stratified for x binary covariates is given by:

\[ T = \sum_{k=1}^{K} w_k d_k \left( \sum_{k=1}^{K} w_k^2 \hat{\sigma}_k^2 \right)^{1/2} \]

where \( K = 2^x \), representing the different cross classifications of each covariate. For example, considering two covariates, we would set \( K=4 \). This test statistic is therefore a weighted combination of the 4 t-test statistics, \( T_k \), within each stratum, with weights:

\[ w_k = \left( \frac{n_{1k} n_{0k}}{n_{1k} + n_{0k}} \right) \left( \sum_{k=1}^{K} \frac{n_{1k} n_{0k}}{n_{1k} + n_{0k}} \right) \]

which are based on the type II sums of squares philosophy. This test statistic is asymptotically distributed as a standard normal.

Note that the t-test statistic in each stratum, \( T_k \), is given by:

\[ T_k = \frac{d_k}{\hat{\sigma}_k} = \frac{(\overline{x}_{1,k} - \overline{x}_{0,k})}{\left( \frac{n_{1k} + n_{0k}}{n_{1k} n_{0k}} \right)^{1/2} \hat{\sigma}_k} \]
where:

\[ n_{1k} = \text{number of subjects in the Test group in stratum } k; \]
\[ n_{0k} = \text{number of subjects in the Active group in stratum } k; \]

\[ \bar{x}_{1,k} = \frac{1}{n_{1k}} \sum_{j=1}^{n_{1k}} x_{1,j,k} = \text{mean of the observations in the Test group in stratum } k; \]

\[ \bar{x}_{0,k} = \frac{1}{n_{0k}} \sum_{j=1}^{n_{0k}} x_{0,j,k} = \text{mean of the observations in the Active group in stratum } k; \]

\[ s_k^2 = \text{pooled estimate of the variance within stratum } k \text{ for the test and active groups.} \]

Using these weights, the stratified estimates of treatment effect and 95% confidence intervals are obtained as follows:

Treatment effect \( d_w = \sum_{k=1}^{K} w_k d_k \)

95% Confidence Interval for \( d_w = d_w \pm Z_{0.025} \sqrt{\sum_{k=1}^{K} w_k^2 \hat{\sigma}_k^2} \)

where \( Z_{0.025} = 1.960 \) is the standard normal deviate for one-sided \( \alpha = 0.025 \)
PROGRAMMING

MACRO FOR PROVISION OF RESULTS (STRAT_T)
The following code can be used in SAS to derive the stratified t-test.
Assumptions are such that the covariates have already been converted to a single STRATA variable; that is, 2^k levels. The input variable of interest is normally distributed.

```sas
%MACRO STRAT_T (datain=, dataout=, outfile=, varint=, treat=, strata=, alpha=, direct=);

options nofmterr nocenter mprint mlogic symbolgen;

** Set defaults **;
%if %length(&ALPHA)<0 %then %let ALPHA=95;
%if %length(&DIRECT)<0 %then %let DIRECT=1;

proc sort data=&datain out=strtdata;
by &strata &treat;
run;

data strtdat2;
set strtdata;
if &strata ne . and &varint ne . and &treat ne . ;
strata=&strata;
treat=&treat;
run;

proc sql;
create table n as select count(&varint) as n,
mean(&varint) as mean, strata, treat from strtdat2
group by strata, treat;
select max(strata) into : stratnum from strtdat2;
quit;
run;

%do str=1 %to &STRATNUM;
%do trt=1 %to 2;
data _null_;
set n;
if strata=&str and treat=&trt then do;
call symput("n&str&str",n);
call symput("mean&str&str",mean);
end;
run;
%end;
%end;
```

** Macro name: STRAT_T.sas **
** Variables : **
** &DATAIN - full dataset name (including libname) **
** &DATAOUT - full output dataset name (including libname) **
** &OUTFILE - full output area (including libname) **
** &VARINT - variable of interest **
** &STRATA - strata variables (derived from binary covars) **
** &ALPHA - confidence level (default=95) **
** &DIRECT - "direction" of test (default=1) **
** Function : Provides stratified t-test and formalised output **

PhUSE 2007
** Set up weights **;

data wt1;
  %do str=1 %to &STRATNUM;
    %do trt=1 %to 2;
      n&trt&str=&&n&trt&str;
      mean&trt&str=&&mean&trt&str;
    %end;
  %end;
  %do str=1 %to &STRATNUM;
    num&str=(&&n1&str*&&n2&str) / (&&n1&str+&&n2&str);
    diff&str=&&mean1&str - &&mean2&str;
  %end;
denom = num1;
  %do str=2 %to &STRATNUM;
    denom = denom+num&str;
  %end;
  %do str=1 %to &STRATNUM;
    wt&str = num&str / denom;
  %end;
est = wt1 * diff1;
  %do str=2 %to &STRATNUM;
    est = est + (wt&str*diff&str);
  %end;
run;

** Set up GLM **;
proc sort data=strtdat2 out=glm;
  by strata;
run;
ods output overallanova=anova;
proc glm data=glm;
  class treat;
  model &varint=treat;
  by strata;
quit;
run;
ods output close;

data error(drop=source);
  set anova;
  where upcase(source)='ERROR';
  var=ms;
run;

data _null_; 
  set error;
  %do str=1 %to &STRATNUM;
    if strata=&str then call symput("error&str",var);
  %end;
run;
** Bring together **;
data &DATAOUT;
set wt1;
%do str=1 %to &STRATNUM;
error&str=&error&str;
var&str=&error&str/num&str;
%end;
varval=(var1*(wt1*wt1));
%do str=2 %to &STRATNUM;
varval=varval + (var&str*(wt&str*wt&str));
%end;
stderr=sqrt(varval);
tval=1 - (((1-(&ALPHA/100)))/2);
low =est-(probit(tval)*stderr);
high=est+(probit(tval)*stderr);
t=est/stderr;
%if &DIRECT=1 %then %do; p=probnorm(t); %end;
%if &DIRECT^=1 %then %do; p=1-probnorm(t); %end;
run;

** Sort and present data in table format **;
data table;
set &DATAOUT;
keep meanse pval ci;
meanse=put(est,5.2)||'  ('||put(stderr,6.3)||')';
pval  =put(p,6.4);
ci    ='('||put(low,6.2)||', '||put(high,6.2)||')';
run;
proc transpose data=table out=table2;
var meanse pval ci;
run;
data table3 (drop=_name_ rename=(col1=outvar));
length subord $20.;
set table2;
order=_n_;
if order=1 then subord="Mean (SE)  ";
if order=2 then subord="p-Value    ";
if order=3 then subord="&ALPHA % CI";
run;
proc sort data=table3 out=finalt;
by order;
run;

** Produce output using PROC REPORT **;
proc printto file="&outfile" new;
run;
title "Stratified t-test output";
proc report data=finalt nowindows spacing=0
    missing headline headskip split='*' ls=160 ps=60;
column order subord ('Treatment Comparison' outvar);
define order   / order order=internal n=0;
define subord  / display flow left width=40 'Statistic';
define outvar  / display center    width=40 ' ';
run;
proc printto;
run;
%MEND;
PhUSE 2007

SAMPLE OUTPUT
Using some simulated data, we have the following output from the above macro:

PROC REPORT output:

<table>
<thead>
<tr>
<th>Statistic</th>
<th>Treatment Comparison</th>
</tr>
</thead>
<tbody>
<tr>
<td>Mean (SE)</td>
<td>-14.1 (3.891)</td>
</tr>
<tr>
<td>p-Value</td>
<td>0.0001</td>
</tr>
<tr>
<td>95% CI</td>
<td>(-21.75, -6.50)</td>
</tr>
</tbody>
</table>

&DATASET dataset (TRANSPOSEd for ease of review):

<table>
<thead>
<tr>
<th>STRATA</th>
<th>TreatA (N)</th>
<th>TreatB (N)</th>
<th>Weights</th>
<th>Treatment Difference</th>
<th>Weighted Treatment Difference</th>
<th>Variance of Difference</th>
<th>Variance of Weighted Difference</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>10</td>
<td>8</td>
<td>0.24845</td>
<td>-18.21167</td>
<td>-4.524638</td>
<td>74.677123</td>
<td>18.553323</td>
</tr>
<tr>
<td>2</td>
<td>9</td>
<td>9</td>
<td>0.25155</td>
<td>-16.55556</td>
<td>-4.164596</td>
<td>49.150202</td>
<td>12.363971</td>
</tr>
<tr>
<td>3</td>
<td>10</td>
<td>8</td>
<td>0.24845</td>
<td>-7.956667</td>
<td>-1.976812</td>
<td>54.368783</td>
<td>13.507772</td>
</tr>
<tr>
<td>4</td>
<td>9</td>
<td>9</td>
<td>0.25155</td>
<td>-13.75926</td>
<td>-13.79526</td>
<td>64.2806</td>
<td>16.169965</td>
</tr>
</tbody>
</table>

LIMITATIONS AND COMPARISONS
Using some simulated data, we can plot the estimates and standard errors against the number of observations from the stratified t-test against the ANCOVA approach.

The simulation simply used the RANUNI function as multiples of variables in order to provide randomized observations within an existing group. The following table represents the number of observations (N) and the estimates and standard errors utilizing the stratified t-test and ANCOVA respectively:

<table>
<thead>
<tr>
<th>N</th>
<th>Stratified t-test</th>
<th>ANCOVA</th>
</tr>
</thead>
<tbody>
<tr>
<td></td>
<td>ESTIMATE</td>
<td>SE</td>
</tr>
<tr>
<td>80</td>
<td>-8.8959</td>
<td>6.221243</td>
</tr>
<tr>
<td>160</td>
<td>-5.60174</td>
<td>4.093138</td>
</tr>
<tr>
<td>320</td>
<td>-3.52742</td>
<td>2.391905</td>
</tr>
<tr>
<td>640</td>
<td>-2.22121</td>
<td>1.332119</td>
</tr>
<tr>
<td>1280</td>
<td>-1.39869</td>
<td>0.723148</td>
</tr>
<tr>
<td>2560</td>
<td>-0.88076</td>
<td>0.386529</td>
</tr>
<tr>
<td>5120</td>
<td>-0.55461</td>
<td>0.204532</td>
</tr>
<tr>
<td>10240</td>
<td>-0.34924</td>
<td>0.107491</td>
</tr>
<tr>
<td>20480</td>
<td>-0.21992</td>
<td>0.056222</td>
</tr>
<tr>
<td>40960</td>
<td>-0.13848</td>
<td>0.029307</td>
</tr>
</tbody>
</table>

One can see from the comparison of estimates, there is no difference between the two approaches; however, looking at the standard errors, there is a slight difference (to 4dp) where N<10000; from this point onwards, the law of large numbers (central limit theorem) is evident.
CONCLUSION
We have examined the methodology behind the stratified t-test and have provided a basic SAS macro to derive the results from the analysis and create a basic table, as well as a more in-depth dataset containing the relevant aspects of the by-stratum criteria.

As with all statistical techniques, we need to examine the data being analyzed before coming to a decision about the methodological approach – although the stratified t-test is an acceptable approach, it is fair to utilize other techniques to provide the same inference.

The simulation provided above shows that there is little difference between the stratified t-test approach and standard analysis of covariance – computational power aside, the stratified t-test will provide a more detailed breakdown of the strata, whereas the ANCOVA approach will provide simple summary statistics for general review.

REFERENCES
2. SJ Senn (1997); Statistical Issues in Drug Development

ACKNOWLEDGMENTS
Thanks to Katherine Hutchinson of Oxford Pharmaceutical Sciences Ltd. for her guidance and patience.

CONTACT INFORMATION
Your comments and questions are valued and encouraged. Contact the author at:
John Salter
Oxford Pharmaceutical Sciences Ltd.
Lancaster House
Kingston Business Park
Kingston Bagpuize
OXFORDSHIRE
OX13 5FE
Work Phone: 01865 823823
Email: john.salter@ops-web.com
Web: www.ops-web.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.